Derivation of scattering from Thin Disks

Thin circular disk

In[1]:=
```
$Assumptions := {R > 0, q > 0, r > 0, Sin[θq] ≥ 0, θq > 0, θq ≤ π/2}
```

Lets assume the disk is in the xy plane, then any scatterer point is given in polar coordinates by r [0:R], ϕ in [-π:π]

In[2]:=
```
Rvec := {r Cos[ϕ], r Sin[ϕ], 0}
```

where the measure is r dr dϕ when integrating in polar coordinates.
The q vector can hit the disk from any direction on a sphere, but due to symmetry we can put the qvector in upper xz plane, using the angle θp that the q vector makes with the z axis. The proper measure is d cos[θp] dϕp = Sin[θq] dθq dϕp. Due to rotation symmetry around z, the ϕq integral gives a 2π constant, that is cancelled by the corresponding normalization constant, and hence can be neglected.

In[3]:=
```
qvec := {q Sin[θq], 0, q Cos[θq]}
```

In[4]:=
```
Rvec.qvec
```

Out[4]= q r Cos[ϕ] Sin[θq]

To normalize, we need this integral, which happens to be the area of the disk:

In[∘]:=
```
Integrate[r, {r, 0, R}, {ϕ, -π, π}]
```

Out[∘]= πR^2

Then form factor amplitude relative to the centre of the disk for a given FIXED q vector is: (where r comes from the measure in polar coordinates, and πR^2 from the normalization.

In[5]:=
```
Aqvec = Integrate[Exp[-I Rvec.qvec] r/(π R^2), {r, 0, R}, {ϕ, -π, π}] // FunctionExpand
```

Out[5]= $\frac{2\,\mathrm{BesselJ}[1,\ q\,R\,\mathrm{Sin}[\theta q]]\,\mathrm{Csc}[\theta q]}{q\,R}$

This expression is for a FIXED q vector that makes an angle of θq with the z axis. Hence we need to average over all the incoming q vectors, the measure is Sin[θq] dθq and we integrate from [0:π/2] corresponding from the north pole (z axis) down to the equator: (this average contribute a normalization constant of 1)

In[∘]:= `Integrate[Sin[θq], {θq, 0, π / 2}]`

Out[∘]= 1

In[6]:= `Adiskcenter = Integrate[Aqvec Sin[θq], {θq, 0, π / 2}]`

Out[6]= $$\int_0^{\frac{\pi}{2}} \frac{2\,\text{BesselJ}[1,\, q\,R\,\text{Sin}[\theta q]]}{q\,R}\, d\theta q$$

In[7]:= `Adiskcenter /. q → x / R`

Out[7]= $$\int_0^{\frac{\pi}{2}} \frac{2\,\text{BesselJ}[1,\, x\,\text{Sin}[\theta q]]}{x}\, d\theta q$$

There is no closed analytic form for this form factor amplitude.
To calculate the form factor of a disk, we recognize that the vector between any two scatterers a,b R=Rb-Ra in the disk, can be written as R=Rb-0-(Ra-0), where 0 is the centre of the disk. Hence the pair distance between the two scatterers is the convolution of two radial distance distributions. Aqvec above is the Fourier transformation this radial distribution for fixed geometry. Fourier transforming the convolution of two identical distributions becomes the square of the Fourier transform Fqvec=Aqvec^2, which is the form factor of a disk in fixed geometry. What remains is to perform the orientational average of the angle between the q vector and the disk:

In[8]:= `Fdisk = Integrate[Aqvec^2 Sin[θq], {θq, 0, π / 2}]`

Out[8]= $$\frac{2\,q\,R - 2\,\text{BesselJ}[1,\, 2\,q\,R]}{q^3\,R^3}$$

In[9]:= `Fdisk /. q → x / R`

Out[9]= $$\frac{2\,x - 2\,\text{BesselJ}[1,\, 2\,x]}{x^3}$$

Reference for form factor of thin disk : O . Kratky and G . Porod, J . Colloid . Sci . 4, 35 (1949) .

Distributed surface reference point .

The form factor amplitude of a scatterer on the disk relative to a randomly chosen point on the disk is identical to the form factor above, similar the phase factor between two randomly chosen points.

Circular rim

A special case of the disk, is where we just consider the rim, which forms a circle.
First the normalization constant, which is its circumference.

In[∘]:= `Integrate[R, {ϕ, -π, π}]`

Out[∘]= $2\,\pi\,R$

Form factor amplitude of circular rim in fixed geometry:

In[10]:=
```
Arimqvec = Integrate[Exp[-I Rvec.qvec] R/(2 π R) /. r → R, {ϕ, -π, π}]
```

Out[10]=
```
BesselJ[0, q R Sin[θq]]
```

Making an orientational average of a circle produces the form factor amplitude of a spherical shell:

In[11]:=
```
Psicenter2rim = Integrate[Arimqvec Sin[θq], {θq, 0, π / 2}] // FunctionExpand
```

Out[11]= $\frac{\text{Sin}[q\,R]}{q\,R}$

With this we can now calculate the form factor of a circular rim, which is the same as the form factor amplitude of the rim relative to a random point, and the phase factor between two randomly chosen points on the rim :

In[12]:=
```
Psirim2rim = Integrate[(Arimqvec)^2 Sin[θq], {θq, 0, π / 2}] // Expand
```

Out[12]= $\text{BesselJ}[0,\, 2\,q\,R] + \frac{1}{2}\,\pi\,\text{BesselJ}[1,\, 2\,q\,R]\,\text{StruveH}[0,\, 2\,q\,R] - \frac{1}{2}\,\pi\,\text{BesselJ}[0,\, 2\,q\,R]\,\text{StruveH}[1,\, 2\,q\,R]$

We can also derive the form factor amplitude of a random scatterer on a disk relative to a random point on its rim, which is also the rim-to-disk surface phase factor :

In[13]:=
```
Adiskrim = Integrate[Aqvec Arimqvec Sin[θq], {θq, 0, π / 2}] // Expand
```

Out[13]= $\int_0^{\frac{\pi}{2}} \frac{2\,\text{BesselJ}[0,\, q\,R\,\text{Sin}[\theta q]]\,\text{BesselJ}[1,\, q\,R\,\text{Sin}[\theta q]]}{q\,R}\, d\theta q$

Not many expression here that are easy to evaluate numerically. The derivation reproduce results of C. Svaneborg and J.S. Pedersen "A formalism for scattering of complex composite structures. II. Distributed reference points" J. Chem. Phys. 136, 154907 (2012); doi: 10.1063/1.3701737

Plots

In[14]:=
```
LogLogPlot[{Fdisk, Adiskcenter, Adiskrim} /. q → x / R // Evaluate, {x, 0.1, 20},
 PlotLabel → "Disk scattering terms", AxesLabel → {"x", "F(x),A(x),Psi(x)"},
 PlotLegends → {"Fdisk", "Frim", "Acentre", "Arimdisk"}, PlotRange → {0.0001, 1}]
```

Out[14]=



Guinier - expansion

Above we derived the mean - square distances explicitly. Here we show how to obtain these from a Guinier expansion of the various scattering terms.

The Guinier expansion of F is

$1 - q^2\,\mathrm{Rg}^2 / 3 + O(q^4)$

In[∘]:=
```
Series[F, {q, 0, 3}]
```

Out[∘]= $1 - \frac{R^2 q^2}{6} + O[q]^4$

Hence to isolate the radius of gyration in the series:

In[∘]:=
```
Series[Fdisk, {q, 0, 3}]
Solve[Normal[%] == 1 - Rg2 q^2/3, Rg2] // Simplify
```

Out[∘]= $1 - \frac{R^2 q^2}{6} + O[q]^4$

Out[∘]= $\left\{\left\{\mathrm{Rg2} \to \frac{R^2}{2}\right\}\right\}$

In[∘]:=
```
Series[Fdisk, {q, 0, 3}]
Solve[Normal[%] == 1 - σR2 q^2/6, σR2] // Simplify
```

Out[∘]= $1-\frac{R^2 q^2}{6}+O[q]^4$

Out[∘]= $\{\{\sigma R2 \to R^2\}\}$

In[∘]:=
```
Series[Adiskcenter, {q, 0, 3}]
Solve[Normal[%] == 1 - σR2 q^2/6, σR2] // Simplify
```

Out[∘]= $1-\frac{R^2 q^2}{12}+O[q]^4$

Out[∘]= $\left\{\left\{\sigma R2 \to \frac{R^2}{2}\right\}\right\}$

Save example data to file for Validation :

In[15]:=
```
Clear[PARENTDIR, DIR1, DIR01]
PARENTDIR = Directory[]
DIR1 := PARENTDIR <> "/Sampled/ThinDisk_R1/"
DIR01 := PARENTDIR <> "/../Examples/Validation/ThinDisk_R1/"
CreateDirectory[DIR01];
```

Out[16]= /home/zqex/source/SEB/Mathematica

CreateDirectory: /home/zqex/source/SEB/Examples/Validation/ThinDisk_R1/ already exists.

In[26]:=
```
SaveFunction[func_, filename_, NN_, qmin_, qmax_] := Module[{}, Export[filename,
    {#, N[func[#]]} & /@ Table[10^(Log[10, qmax/qmin] * i/NN + Log[10, qmin]), {i, 0, NN}]]]
SetAttributes[SaveFunction, HoldAll]
```

In[28]:=
```
Clear[qvec, qq]
qvec[qmin_, qmax_, NN_] :=
 Table[10^(Log[10, qmax/qmin] * i/NN + Log[10, qmin]), {i, 0, NN}]
qq := qvec[0.8, 50, 500] // N
```

Form factor :

In[∘]:=
```
Clear[Term, Func1, DATA]
Term[q_] = Fdisk
Func1[q_] := Term[q] /. R → 1
FILE = "F.q";
OFILE = DIR01 <> "FF.dat"
SaveFunction[Func1, OFILE, 200, 0.01, 50];
DATA = {#[[1]] , Abs[#[[2]]]} & /@ Delete[Import[DIR1 <> FILE, "Table"], 1];
ListLogLogPlot[{DATA, {# , Abs[Func1[#]]} & /@ qq},
 PlotStyle → {{Red, Thick}, Black}, Joined → {False, True}]
```

Out[∘]=
$$\frac{2\,q\,R - 2\,\mathrm{BesselJ}[1,\,2\,q\,R]}{q^3\,R^3}$$

Out[∘]= /home/zqex/source/SEB/Mathematica/../Examples/Validation/ThinDisk_R1/FF.dat

Out[∘]=



Form factor amplitude (center):

In[∘]:=
```
Clear[Term, Func1, DATA]
Term[q_] = Adiskcenter
Func1[q_] := Term[q] /. R → 1
FILE = "FFA_center.q";
OFILE = DIR01 <> "FFA_center.dat"
SaveFunction[Func1, OFILE, 200, 0.01, 50];
DATA = {#[[1]] , Abs[#[[2]]]} & /@ Delete[Import[DIR1 <> FILE, "Table"], 1];
ListLogLogPlot[{DATA, {# , Abs[Func1[#]]} & /@ qq},
 PlotStyle → {{Red, Thick}, Black}, Joined → {False, True}]
```

Out[∘]=
$$\int_0^{\frac{\pi}{2}} \frac{2\,\mathrm{BesselJ}[1,\ q\,R\,\mathrm{Sin}[\theta q]]}{q\,R}\,d\theta q$$

Out[∘]= /home/zqex/source/SEB/Mathematica/../Examples/Validation/ThinDisk_R1/FFA_center.dat

NIntegrate: The precision of the argument function (2.5 BesselJ[1, 0.8 Sin[θq]]) is less than WorkingPrecision (26.`).

NIntegrate: The precision of the argument function (2.47941 BesselJ[1, 0.806644 Sin[θq]]) is less than WorkingPrecision (26.`).

NIntegrate: The precision of the argument function (2.45899 BesselJ[1, 0.813343 Sin[θq]]) is less than WorkingPrecision (26.`).

General: Further output of NIntegrate::precw will be suppressed during this calculation.

Out[∘]=


Form factor amplitude (rim):

In[∘]:=
```
Adiskrim
```

Out[∘]=
$$\int_0^{\frac{\pi}{2}} \frac{2\,\text{BesselJ}[0,\, q\,R\,\text{Sin}[\theta q]]\,\text{BesselJ}[1,\, q\,R\,\text{Sin}[\theta q]]}{q\,R}\,d\theta q$$

In[∘]:=
```
{#, Abs[Func1[#]]} & /@ qq
```

Out[∘]=
```
{q Sin[θq], 0., q Cos[θq]}[{0.8, 0.806138}, {50., 0.00922663}, {500., 0.0010047}]
```

In[∘]:=
```
Clear[Term, Func1, DATA]
Term[q_] = Adiskrim
Func1[q_] := Term[q] /. R → 1 /. r → 1 // N
FILE = "FFA_rim.q";
OFILE = DIRO1 <> "FFA_rim.dat"
SaveFunction[Func1, OFILE, 200, 0.01, 50];
DATA = {#[[1]], Abs[#[[2]]]} & /@ Delete[Import[DIR1 <> FILE, "Table"], 1];
ListLogLogPlot[{DATA, {#, Abs[Func1[#]]} & /@ qq},
 PlotStyle → {{Red, Thick}, Black}, Joined → {False, True}]
```

Out[∘]=
$$\int_0^{\frac{\pi}{2}} \frac{2\,\text{BesselJ}[0,\, q\,R\,\text{Sin}[\theta q]]\,\text{BesselJ}[1,\, q\,R\,\text{Sin}[\theta q]]}{q\,R}\,d\theta q$$

Out[∘]= /home/zqex/source/SEB/Mathematica/../Examples/Validation/ThinDisk_R1/FFA_rim.dat

Out[∘]=



Phase factor rim–to–rim:

In[∘]:=
```
Clear[Term, Func1, DATA]
Term[q_] = Psirim2rim
Func1[q_] := Term[q] /. R → 1 /. r → 1
FILE = "P_rim2rim.q";
OFILE = DIRO1 <> "Psi_rim2rim.dat"
SaveFunction[Func1, OFILE, 200, 0.01, 50];
DATA = {#[[1]], Abs[#[[2]]]} & /@ Delete[Import[DIR1 <> FILE, "Table"], 1];
ListLogLogPlot[{DATA, {#, Abs[Func1[#]]} & /@ qq},
 PlotStyle → {{Red, Thick}, Black}, Joined → {False, True}]
```

Out[∘]= $\text{BesselJ}[0, 2 q R] + \frac{1}{2} \pi\, \text{BesselJ}[1, 2 q R]\, \text{StruveH}[0, 2 q R] - \frac{1}{2} \pi\, \text{BesselJ}[0, 2 q R]\, \text{StruveH}[1, 2 q R]$

Out[∘]= /home/zqex/source/SEB/Mathematica/../Examples/Validation/ThinDisk_R1/Psi_rim2rim.dat

Out[∘]=


Phase factor center–to–rim:

In[31]:=
```
Clear[Term, Func1, DATA]
Term[q_] = Psicenter2rim
Func1[q_] := Term[q] /. R → 1 /. r → 1
OFILE = DIR01 <> "Psi_center2rim.dat"
SaveFunction[Func1, OFILE, 200, 0.01, 50];
ListLogLogPlot[{{# , Abs[Func1[#]]} & /@ qq},
 PlotStyle → {{Red, Thick}, Black}, Joined → {False, True}]
```

Out[32]= $\frac{\mathrm{Sin}[q\,R]}{q\,R}$

Out[34]= /home/zqex/source/SEB/Mathematica/../Examples/Validation/ThinDisk_R1/Psi_center2rim.dat

Out[36]=



Size estimations:

In[∘]:=
```
Series[Fdisk, {q, 0, 3}];
Solve[Normal[%] == 1 - Rg2 q^2/3, Rg2] // Simplify;
Rg2 /. %[[1]]
```

Out[∘]= $\frac{R^2}{2}$

In[∘]:=
```
Series[Adiskcenter, {q, 0, 3}];
Solve[Normal[%] == 1 - sigmaR2 q^2/6, sigmaR2] // Simplify;
sigmaR2 /. %[[1]]
```

Out[∘]= $\frac{R^2}{2}$

In[•]:= $\texttt{Series[Fdisk, \{q, 0, 3\}];}$

$\texttt{Solve}\left[\texttt{Normal[\%] == 1} - \dfrac{\texttt{sigmaR2 q}^2}{6}, \texttt{sigmaR2}\right] \texttt{// Simplify;}$

$\texttt{sigmaR2 /. \%[[1]]}$

Out[•]= R^2

In[•]:= $\texttt{Arim}$

Out[•]= $\displaystyle\int_0^{\frac{\pi}{2}} \frac{2\,\text{BesselJ}[0, x\,\text{Sin}[\theta q]]\,\text{BesselJ}[1, x\,\text{Sin}[\theta q]]}{x}\, d\theta q$

In[•]:= $\texttt{Series}\left[\dfrac{2\,\texttt{BesselJ}[0, x\,\texttt{Sin}[\theta q]]\,\texttt{BesselJ}[1, x\,\texttt{Sin}[\theta q]]}{x} \texttt{ /. } x \to q\,R, \{q, 0, 5\}\right]$

Out[•]= $\text{Sin}[\theta q] - \frac{3}{8}\left(R^2\,\text{Sin}[\theta q]^3\right) q^2 + \frac{5}{96} R^4\,\text{Sin}[\theta q]^5\, q^4 + O[q]^6$

In[•]:= $\texttt{Integrate}\left[\texttt{Sin}[\theta q] - \frac{3}{8}\left(R^2\,\texttt{Sin}[\theta q]^3\right) q^2 + \frac{5}{96} R^4\,\texttt{Sin}[\theta q]^5\, q^4, \{\theta q, 0, \pi/2\}\right] \texttt{// Expand}$

Out[•]= $1 - \dfrac{q^2 R^2}{4} + \dfrac{q^4 R^4}{36}$

$\texttt{Solve}\left[1 - \dfrac{q^2 R^2}{4} \texttt{ == } 1 - \dfrac{\texttt{sigmaR2 q}^2}{6}, \texttt{sigmaR2}\right] \texttt{// Simplify;}$

$\texttt{sigmaR2 /. \%[[1]]}$

Out[•]= $\dfrac{3 R^2}{2}$

In[•]:= $\texttt{Series[Arimcenter, \{q, 0, 3\}];}$

$\texttt{Solve}\left[\texttt{Normal[\%] == 1} - \dfrac{\texttt{sigmaR2 q}^2}{6}, \texttt{sigmaR2}\right] \texttt{// Simplify;}$

$\texttt{sigmaR2 /. \%[[1]]}$

Out[•]= r^2

In[•]:= $\texttt{Series[Psirim2rim, \{q, 0, 3\}];}$

$\texttt{Solve}\left[\texttt{Normal[\%] == 1} - \dfrac{\texttt{sigmaR2 q}^2}{6}, \texttt{sigmaR2}\right] \texttt{// Simplify;}$

$\texttt{sigmaR2 /. \%[[1]]}$

Out[•]= $2 r^2$