Derivation of scattering from a solid sphere.

In[195]:= `$Assumptions := {R > 0, q > 0, r > 0, Sin[θq] ≥ 0, θq > 0, θq ≤ π/2}`

We start by looking at a sphere, which is the easiest solid body to work with given its symmetry . Any scatterer within the solid sphere can be expressed with a spherical coordinate:

In[196]:= `Rvec := {r Cos[θ] Sin[ϕ], r Sin[θ] Sin[ϕ], r Cos[ϕ]}`

Here ϕ is the angle Rvec makes with the z axis and θ is the angle the xy projection makes with the x axis. The measure with this choice of coordinates is $r^2\, d\theta\, d\mathrm{Cos}[\phi]$

We can place the q vecor along any direction due to rotational symmetry, easiest direction is z

In[197]:= `qvec = {0, 0, q}`

Out[197]=

{0, 0, q}

In[198]:= `Rvec.qvec`

Out[198]=

$q\, r\, \mathrm{Cos}[\phi]$

Some useful normalization constants : surface of the unit sphere, and radial factor for calculating the volume of a sphere:

In[199]:= `Integrate[Sin[θ], {θ, 0, π}, {ϕ, -π, π}]`

Out[199]=

4π

In[200]:= `Integrate[r^2, {r, 0, R}]`

Out[200]=

$$\frac{R^3}{3}$$

To get the form factor amplitude relative to the centre of the sphere, we integrate the interference contribution from all scatterers on a spherical surface at r:

In[201]:=
```
Ashell = Integrate[Exp[I q r Cosθ]/(4 π), {Cosθ, -1, 1}, {ϕ, -π, π}]
```

Out[201]=

$$\frac{\text{Sin}[q\, r]}{q\, r}$$

Which is well known from the Debye formula. To get a the form factor amplitude of a sphere, we have an additional integral over r in [0 : R] taking the measure $4\,\pi\, r^2$ and normalization volume $(4\,\pi\, R^3 / 3)$ into account :

In[202]:=
```
Aspherecenter = Integrate[Ashell r^2/(R^3/3), {r, 0, R}]
```

Out[202]=

$$\frac{3\left(-q\, R\, \text{Cos}[q\, R] + \text{Sin}[q\, R]\right)}{q^3\, R^3}$$

Which is the form factor amplitude relative to the centre of the sphere . Any pair distance between two scatterers can be stated as the convolution of two vectors connecting a scatterer to the origin . Fourier transforming the pair distance turns it into products of the Fourier transforms . Hence the form factor is just the form factor amplitude squared :

In[203]:=
```
Fsphere = Aspherecenter^2
```

Out[203]=

$$\frac{9\left(-q\, R\, \text{Cos}[q\, R] + \text{Sin}[q\, R]\right)^2}{q^6\, R^6}$$

Reference to form factor of sphere: L . Rayleigh, Proc . R . Soc ., London . A84, 25 (1911) .

Radius of gyration of a sphere :

In[61]:=
```
Solve[Normal[Series[Fsphere, {q, 0, 3}]] == 1 - Rg2 q^2/3, Rg2]
```

Out[61]=

$$\left\{\left\{\text{Rg2} \to \frac{3\, R^2}{5}\right\}\right\}$$

MSD beween random points inside a sphere . (sigma = 2)

In[ ]:=
```
Solve[Normal[Series[Fsphere, {q, 0, 3}]] == 1 - σR2 q^2/6, σR2]
```

Out[ ]=

$$\left\{\left\{\sigma\text{R2} \to \frac{6\, R^2}{5}\right\}\right\}$$

MSD from the centre to any scatterer in the sphere (sigma=1):

In[∘]:=
```
Solve[Normal[Series[Aspherecenter, {q, 0, 3}]] == 1 - σR2 q^2/6, σR2]
```
Out[∘]= $\left\{\left\{\sigma R2 \to \frac{3\ R^2}{5}\right\}\right\}$

Since the sphere was derived from integrating radial shells, we can also inverse this calculation and rederive the form factor amplitude of the shell by differentiating wrt . R when proper volume and area measures are taken into account .

In[∘]:=
```
D[4 π R^3/3 Aspherecenter, R] / (4 π R^2) // Simplify
```
Out[∘]= $\frac{\mathrm{Sin}[q\ R]}{q\ R}$

The form factor of a spherical shell is

In[204]:=
```
Fshell = Ashell^2
```
Out[204]= $\frac{\mathrm{Sin}[q\ r]^2}{q^2\ r^2}$

MSD from a random site on the surface to another random site on the surface (sigma=2)

In[∘]:=
```
Solve[Normal[Series[Fshell, {q, 0, 3}]] == 1 - Rg2 q^2/3, Rg2]
```
Out[∘]= $\{\{Rg2 \to r^2\}\}$

In[∘]:=
```
Solve[Normal[Series[Fshell, {q, 0, 3}]] == 1 - σR2 q^2/6, σR2]
```
Out[∘]= $\{\{\sigma R2 \to 2\ r^2\}\}$

By construction same result as for Rg2=R2 since sigma=2.

MSD from center to surface . For a sphere this is R by definition . NB sigma = 1.

In[∘]:=
```
Solve[Normal[Series[Ashell, {q, 0, 3}]] == 1 - σR2 q^2/6, σR2]
```
Out[∘]= $\{\{\sigma R2 \to r^2\}\}$

The form factor amplitude of a sphere relative to a random point on the surface, is the convolution of a step from the surface to the centre and from the centre to a site inside this sphere:

In[205]:=
```
Asphereshell = Ashell Aspherecenter /. r → R
```
Out[205]= $\frac{3\ \mathrm{Sin}[q\ R]\ (-q\ R\ \mathrm{Cos}[q\ R] + \mathrm{Sin}[q\ R])}{q^4\ R^4}$

MSD from random point on surface to random point inside the sphere (sigma=1):

In[∘]:=
```
Solve[Normal[Series[Asphereshell, {q, 0, 3}]] == 1 - σR2 q^2/6, σR2]
```

$$\text{Solve}\left[\text{Normal}[\text{Series}[\text{Asphereshell}, \{q, 0, 3\}]] == 1 - \frac{\sigma R2\, q^2}{6}, \sigma R2\right]$$

Out[∘]=

$$\left\{\left\{\sigma R2 \to \frac{8\, R^2}{5}\right\}\right\}$$

Comparing to sampled data and saving data for validation:

In[206]:=
```
Clear[PARENTDIR, DIR1, DIRO1]
```

In[207]:=
```
PARENTDIR = Directory[]
```

Out[207]=

/home/zqex/source/SEB/Mathematica

In[208]:=
```
DIR1 := PARENTDIR <> "/Sampled/SolidSphere_R1.000000/"
DIRO1 := PARENTDIR <> "/../Examples/Validation/SolidSphere_R1/"
CreateDirectory[DIRO1];
```

CreateDirectory: /home/zqex/source/SEB/Examples/Validation/SolidSphere_R1/ already exists.

In[211]:=
```
SaveFunction[func_, filename_, NN_, qmin_, qmax_] := Module[{}, Export[filename,
   {#, N[func[#]]} & /@ Table[10^(Log[10, qmax / qmin] * i / NN + Log[10, qmin]), {i, 0, NN}]]]
SetAttributes[SaveFunction, HoldAll]
```

In[213]:=
```
Clear[qvec, qq]
```

In[214]:=
```
qvec[qmin_, qmax_, NN_] :=
  Table[10^(Log[10, qmax / qmin] * i / NN + Log[10, qmin]), {i, 0, NN}]
qq := qvec[0.8, 50, 500] // N
```

Form factor :

In[327]:=
```
Clear[Term, Func1, DATA]
Term[q_] = Fsphere
Func1[q_] := Term[q] /. R → 1
FILE = "FF.q";
OFILE = DIRO1 <> "F.dat"
SaveFunction[Func1, OFILE, 200, 0.01, 50];
DATA = {#[[1]] , Abs[#[[2]]]} & /@ Delete[Import[DIR1 <> FILE, "Table"], 1];
ListLogLogPlot[{DATA, {# , Abs[Func1[#]]} & /@ qq},
 PlotStyle → {{Red, Thick}, Black}, Joined → {False, True}]
```

Out[328]=

$$\frac{9\left(-q\,R\,\mathrm{Cos}[q\,R]+\mathrm{Sin}[q\,R]\right)^2}{q^6\,R^6}$$

Out[331]=

/home/zqex/source/SEB/Mathematica/../Examples/Validation/SolidSphere_R1/F.dat

Out[334]=



Form factor amplitude (center) :

In[335]:=
```
Clear[Term, Func1, DATA, FILE, OFILE]
Term[q_] = Asherecenter
Func1[q_] := Term[q] /. R → 1
FILE = "FFA_center.q";
OFILE = DIRO1 <> "FFA_center.dat"
SaveFunction[Func1, OFILE, 200, 0.01, 50];
DATA = {#[[1]], Abs[#[[2]]]} & /@ Delete[Import[DIR1 <> FILE, "Table"], 1];
ListLogLogPlot[{DATA, {#, Abs[Func1[#]]} & /@ qq},
 PlotStyle → {{Red, Thick}, Black}, Joined → {False, True}]
```

Out[336]=

$$\frac{3\left(-q\,R\,\mathrm{Cos}[q\,R]+\mathrm{Sin}[q\,R]\right)}{q^3\,R^3}$$

Out[339]=

/home/zqex/source/SEB/Mathematica/../Examples/Validation/SolidSphere_R1/FFA_center.dat

Out[342]=



Form factor amplitude relative to surface:

In[343]:=
```
Clear[Term, Func1, DATA, FILE, OFILE]
Term[q_] = Asphereshell
Func1[q_] := Term[q] /. R → 1
FILE = "FFA_surface.q";
OFILE = DIR01 <> "FFA_surface.dat"
SaveFunction[Func1, OFILE, 200, 0.01, 50];
DATA = {#[[1]] , Abs[#[[2]]]} & /@ Delete[Import[DIR1 <> FILE, "Table"], 1];
ListLogLogPlot[{DATA, {# , Abs[Func1[#]]} & /@ qq},
 PlotStyle → {{Red, Thick}, Black}, Joined → {False, True}]
```

Out[344]=

$$\frac{3\,\mathrm{Sin}[q\,R]\left(-q\,R\,\mathrm{Cos}[q\,R]+\mathrm{Sin}[q\,R]\right)}{q^4\,R^4}$$

Out[347]=

/home/zqex/source/SEB/Mathematica/../Examples/Validation/SolidSphere_R1/FFA_surface.dat

Out[350]=



Phase factor relative to center-to-surface:

In[351]:=
```
Clear[Term, Func1, DATA, FILE, OFILE]
Term[q_] = Ashell
Func1[q_] := Term[q] /. r → 1
FILE = "PF_center_surface.q";
OFILE = DIR01 <> "PF_center_surface.dat"
SaveFunction[Func1, OFILE, 200, 0.01, 50];
DATA = {#[[1]] , Abs[#[[2]]]} & /@ Delete[Import[DIR1 <> FILE, "Table"], 1];
ListLogLogPlot[{DATA, {# , Abs[Func1[#]]} & /@ qq},
 PlotStyle → {{Red, Thick}, Black}, Joined → {False, True}]
```

Out[352]=

$$\frac{\sin[q\, r]}{q\, r}$$

Out[355]=

```
/home/zqex/source/SEB/Mathematica/../Examples/Validation/SolidSphere_R1/PF_center
  _surface.dat
```

Out[358]=



Phase factor surface-to-surface:

In[359]:=
```
Clear[Term, Func1, DATA, FILE, OFILE]
Term[q_] = Fshell
Func1[q_] := Term[q] /. r → 1
FILE = "PF_surface_surface.q";
OFILE = DIR01 <> "PF_surface_surface.dat"
SaveFunction[Func1, OFILE, 200, 0.01, 50];
DATA = {#[[1]] , Abs[#[[2]]]} & /@ Delete[Import[DIR1 <> FILE, "Table"], 1];
ListLogLogPlot[{DATA, {# , Abs[Func1[#]]} & /@ qq},
 PlotStyle → {{Red, Thick}, Black}, Joined → {False, True}]
```

Out[360]=

$$\frac{\mathrm{Sin}[q\,r]^2}{q^2\,r^2}$$

Out[363]=

/home/zqex/source/SEB/Mathematica/../Examples/Validation/SolidSphere_R1/PF_surface_surface.dat

Out[366]=